und dem Zeitbereich (im Plot) spielen (Abb. 3.12).

```
> w1:=2:w2:=10:
> #w1:=1: w2:=1.000001: # ideal
> plot(evalc(Re(f)),t=2..15);
```



Abb. 3.12: Schwingungskurve zu einem Frequenzband

3.1.4 Gaußverteilung und Resonanzlinien

Daß die Fouriertransformation ein wichtiges mathematisches Hilfsmittel in der Schwingungs- und Wellenphysik ist, sieht man daran, daß sie bei solch fundamentalen Aussagen wie der Unschärferelation benötigt wird. Nach der soeben untersuchten idealisierten Rechtecksverteilung wollen wir uns deshalb noch mit natürlichen Verteilungen beschäftigen. Die Gaußverteilung führt letzten Endes auf die Heisenbergsche Unschärferelation, während die Untersuchung exponentiell abklingender Schwingungen mit Hilfe der Fouriertransformation in gleichem Maß für die klassische Physik und die Quantenphysik von Bedeutung ist.

Gaußverteilung

Wir setzen eine Cosinus-Schwingung mit der Gaußfunktion als Einhüllender an und berechnen die Fouriertransformierte

```
> a:='a':
> F:=simplify(fourier(exp(-t^2)*cos(a*t),t,w));
```

$$F := \frac{1}{2}\sqrt{\pi}\,(\,e^{(\,w\,a\,)}+1\,)\,e^{\left(-1/4\,(\,w+a\,)^2\right)}$$

Das Spektrum ist also wieder eine Gaußverteilung (incl. negative Frequenzen), den zugehörigen Plot können Sie im Worksheet erzeugen.

Man kann umgekehrt von einer Gaußverteilung der Frequenzen ausgehen (ohne `assume` wird die Rücktransformation nicht durchgeführt):

```
> restart:with(plots):readlib(fourier):
> assume(sigma>0);
> spek:=exp(-((w-w0)/sigma)^2/2)/sqrt(2*Pi);
> f:=invfourier(spek,w,t);
```

$$spek := \frac{1}{2}\,\frac{e^{\left(-1/2\,\frac{(w-w0)^2}{\sigma\tilde{}^2}\right)}\sqrt{2}}{\sqrt{\pi}}$$

$$f := \frac{1}{2}\,\frac{\sigma\tilde{}\;e^{\left(-1/2\,t^2\,\sigma\tilde{}^2+I\,t\,w0\right)}}{\pi}$$

Eine Gaußlinie erzeugt ein Gaußpaket. Wir stellen beides gemeinsam dar (vgl. Abb. 3.13):

```
> w0:=10: sw:=0.7: ref:=evalc(Re(f)): abf:=evalc(abs(f)):
> psp:=plot(spek,w=0..20): psp;
> pf:=plot({ref,abf,-abf},t=-5..5): pf;
> display({pf,psp});
```



Abb. 3.13: Gaußpaket und zugehöriges Frequenzspektrum

Wenn Sie das Produkt der Varianzen der beiden Gaußfunktionen berechnen, haben Sie die Unschärferelation fast bewiesen.

Resonanzlinien

Natürlich gehört neben dem „Gaußschen Schwingungspaket“ die Lorentz-Linie, also die Untersuchung von Resonanz mit Hilfe der Fouriertransformation noch ins Standardrepertoire:

```
> restart:readlib(fourier):
```

Gedämpfte Schwingung (vgl. Anhang A.2.2, Seite 265, Worksheet *mld2g1.ms)*

```
> damp:=exp((-p/2+I*wurzel)*t);
```

$$damp := \mathrm{e}^{((-1/2\,p + I\,wurzel)\,t)}$$

Fouriertransformierte

```
> assume(q>0,p>0):
> #p:='p':q:='q':
> F:=fourier(damp*Heaviside(t),t,w);
```

$$F := \frac{1}{\frac{1}{2}\,p\tilde{} + I\,(w - wurzel)}$$

Betragsquadrat (mit gleichem Namen)

```
> F:=evalc(abs(F))^2;
```

$$F := 4\,\frac{1}{p\tilde{}^{2} + 4\,w^2 - 8\,w\,wurzel + 4\,wurzel^2}$$

Gängige Darstellung

```
> with(student):
> F:=completesquare(F,w);
```

$$F := 4\,\frac{1}{4\,(w - wurzel)^2 + p\tilde{}^{2}}$$

`assume` loswerden:

```
> F:=subs(p=ph,F): p:='p': F:=subs(ph=p,F);
```

$$F := 4\,\frac{1}{4\,(w - wurzel)^2 + p^2}$$

Darstellung von Resonanzkurven zu verschiedenen Dämpfungen (Abb. 3.14)

```
> q:='q': p:='p':
> wurzel:=sqrt(4*q-p^2):
> q:=10:
> plot({seq(F,p=1..3)},w=0..3*wurzel);
```



Abb. 3.14: Resonanzkurven zu verschiedenen Dämpfungen

Eine weitere Variante der Darstellung der Lorentz-Linie finden Sie im Worksheet (*fourier.ms*). Sie ist typisch für die Atomphysik: Ein angeregter Zustand zerfällt mit der Zerfallskonstanten Γ bzw. hat die mittlere Lebensdauer $\tau = 1/\Gamma$. Dabei wird Strahlung mit der Frequenz ω_0 ausgesandt

$$damp := e^{(-t\,\Gamma)} \cos(\,w0\,t\,) \,\mathrm{Heaviside}(\,t\,)$$

Die Fouriertransformierte ist

$$Lorentz := \frac{1}{4} \frac{1}{\Gamma^2 + (\,w - w0\,)^2}$$

Zur Darstellung (und Übung) kann dies wieder rücktransformiert werden

$$f := \frac{1}{4} \frac{\pi\, e^{(-I\,t\,w0\,)} \left(e^{\left(-\sqrt{\Gamma^2}\,t\right)} \mathrm{Heaviside}(\,t\,) + e^{\left(\sqrt{\Gamma^2}\,t\right)} \mathrm{Heaviside}(\,-t\,)\right)}{\sqrt{\Gamma^2}}$$

Anhand der Plots können Sie nun Folgendes untersuchen: Die Lorentz-Linie fällt für $\Delta\omega = \omega - \omega_0 = \Gamma$ auf den halben Wert, man kann also Γ mit der Linienbreite oder Frequenzunschärfe identifizieren. Andererseits ist $1/\Gamma$ die mittlere Lebensdauer (d.h. die Schwingung klingt in dieser Zeitspanne auf den *e*-ten Teil ab), so daß sich ergibt: $\Delta\omega\tau \sim 1$, also ist die Lebensdauer umgekehrt proportional zur Frequenzunschärfe. Wie schon gesagt: Reine Schwingungen leben am längsten (und benötigen die längste Zeit für ihre Entstehung, trotzdem kann man eine Stimmgabel anschlagen). *fourier.ms*